

SONY

Start guide and tutorial of Spresense programming with Arduino IDE

R&D Center Tokyo Laboratory 14

Kentaro Matsuura

Satoru Iwasaki

Copyright 2020 Sony Corporation

Agenda

- Start guide
 - Install Arduino IDE
 - Install driver for Spresense
 - Install Spresense Arduino Library
 - Connect Spresense to you PC
 - Install Bootloader
 - Write your own “sketch” on Arduino IDE
- Tutorial
 - How to use GNSS in SPRESENSE
 - How to get the disaster information from QZSS in SPRESENSE
 - Play the recoded sound

Start guide

- Install Arduino IDE
- Install driver for Spresense
- Install Spresense Arduino Library
- Connect Spresense to you PC
- Install Bootloader
- Write your own “sketch” on Arduino IDE

Install Arduino IDE



ARDUINO 1.8.9

The open-source Arduino Software (IDE) makes it easy to write code and upload it to the board. It runs on Windows, Mac OS X, and Linux. The environment is written in Java and based on Processing and other open-source software.

This software can be used with any Arduino board. Refer to the [Getting Started](#) page for Installation instructions.

Windows Installer, for Windows XP and up
Windows ZIP file for non admin install

Windows app Requires Win 8.1 or 10
[Get](#) 

Mac OS X 10.8 Mountain Lion or newer


Linux 32 bits
Linux 64 bits
Linux ARM 32 bits
Linux ARM 64 bits

[Release Notes](#)
[Source Code](#)
[Checksums \(sha512\)](#)

<https://www.arduino.cc/en/main/software#>

Install driver for Spresense

https://developer.sony.com/develop/spresense/docs/arduino_set_up_en.html



[CP210x USB to serial driver for Windows 7/8/8.1](#)

[CP210x USB to serial driver \(v10.1.3\) for Windows 10](#)

[CP210x USB to serial driver for Macintosh OSX](#)


If you want to get the latest version, visit [the silicon labs site](#) directly

Install Spresense Arduino Library

https://developer.sony.com/develop/spresense/docs/arduino_set_up_en.html

STEP 1


Open Preferences



STEP 2


Copy and paste the following URL into the field called Additional Boards Managers URLs:

https://github.com/sonydevworld/spresense-arduino-compatible/releases/download/generic/package_spresense_index.json



STEP 3

Open Boards Manager




※ Proxy settings is at "Network" tab on "Preference" window

Install Spresense Arduino Library


https://developer.sony.com/develop/spresense/docs/arduino_set_up_en.html

STEP 4

Search for Spresense and select it and click install




Connect Spresense to you PC



Install Bootloader


STEP 1

Select Spresense board




STEP 2

Select the serial port



STEP 3


Select Spresense Firmware Updater



Install Bootloader


STEP 4

Select "Burn Bootloader"




STEP 5

Review and accept the EULA by following the dialog



STEP 6


Once the upload completes, this is how it should look like



Write your own “sketch” on Arduino IDE

setup function is called once just after system boot

loop function is called periodically during power on



```
1=void setup() {  
2  pinMode(LED0, OUTPUT);  
3  pinMode(LED1, OUTPUT);  
4  pinMode(LED2, OUTPUT);  
5  pinMode(LED3, OUTPUT);  
6 }  
7  
8=void loop() {  
9  digitalWrite(LED0, HIGH);  
10 digitalWrite(LED1, HIGH);  
11 digitalWrite(LED2, HIGH);  
12 digitalWrite(LED3, HIGH);  
13 }
```

pinMode configures input/output mode of a pin. See [Arduino Language Reference](#) for more details

digitalWrite sets HIGH or LOW voltage of a pin. See [Arduino Language Reference](#) for more details

Write your own “sketch” on Arduino IDE


Copy and paste the following example LED test code into a new sketch:

```
void setup() {  
  pinMode(LED0, OUTPUT);  
  pinMode(LED1, OUTPUT);  
  pinMode(LED2, OUTPUT);  
  pinMode(LED3, OUTPUT);  
}  
  
void loop() {  
  digitalWrite(LED0, HIGH);  
  delay(100);  
  digitalWrite(LED1, HIGH);  
  delay(100);  
  digitalWrite(LED2, HIGH);  
  delay(100);  
  digitalWrite(LED3, HIGH);  
  delay(1000);  
  
  digitalWrite(LED0, LOW);  
  delay(100);  
  digitalWrite(LED1, LOW);  
  delay(100);  
  digitalWrite(LED2, LOW);  
  delay(100);  
  digitalWrite(LED3, LOW);  
  delay(1000);  
}
```

Write your own “sketch” on Arduino IDE


STEP 1

Click update button




STEP 2

Wait for the upload to complete



STEP 3


Select baud rate



※You may increase the baud rate to shorten the transfer/upload time. This may not always work depending on what hardware you use (PC, cables etc). A recommendation is to start at the default value and increase the baud rate to test if the transfer still works.

Tutorial 1

– How to use GNSS in SPRESENSE



GNSS (Global Navigation Satellite System)

GPS Chip Antenna




GPS (L1-CA)	GLONASS (L1)	QZSS (L1-C/A)	WAAS	QZSS (L1-S)	95% Accuracy ※	Effective place
○					< 5m	Under the open sky
○		○			< 5m	In East Asia and Oceania
○	○				< 7.8m	In the city
○	○	○			< 7.8m	In the city of East Asia and Oceania
○			○		< 2m	In North America
○		○		○	< 2m	In Japan

※ The positioning error(Accuracy) is a reference value under good conditions. It varies depending on your system and environment

Spresense with external GNSS antenna

Land for uFL connector



Antenna	R29	R31	R33	R30	R32	Note
Chip (on board)	CLOSE	OPEN	CLOSE	OPEN	CLOSE	default
Passive (external)	OPEN	CLOSE	OPEN	OPEN	CLOSE	
Active (external)	CLOSE	OPEN	CLOSE	CLOSE	OPEN	3.3V supplied to antenna



If you want to know further.
Please see the below site.

https://developer.sony.com/develop/spresense/docs/hw_docs_en.html#_how_to_use_the_external_antenna_for_gnss





Open the GNSS sketch

- File → Examples → GNSS → gnss
- You can see the sketch in new window



Monitoring


- Upload
- Tool → Serial monitor



Check the serial monitor


You can see the GNSS Info

- Before positioning is completed
 - Time starts from 1980/01/06, 00:00:00
 - numSat: Number of satellite
 - No-fix and No position are shown
- After positioning is completed
 - correct time, "Fix", latitude and longitude
 - Type: types of satellite, id: satellite number
 - Elv: Elevation angle, Azm: Azimuth,
 - CN0: signal strength are shown



```
COM20 (Spresense)
SpGnss : begin in
SpGnss : begin out
SpGnss : start in
mode = COLD_START
SpGnss : start out
Gnss setup OK
1980/01/06 00:00:01.000500, numSat: 0, No-Fix, No Position
1980/01/06 00:00:02.000522, numSat: 2, No-Fix, No Position
1980/01/06 00:00:03.000513, numSat: 2, No-Fix, No Position
1980/01/06 00:00:04.000505, numSat: 2, No-Fix, No Position
1980/01/06 00:00:05.000496, numSat: 3, No-Fix, No Position
1980/01/06 00:00:06.000518, numSat: 3, No-Fix, No Position
```

Autoscroll ☐ Show timestamp ☐ Newline ☐ 115200 baud ☐ Clear output



```
COM20 (Spresense)
2019/08/27 13:53:58.000569, numSat: 6, Fix, Lat=35.681145, Lon=139.761145
2019/08/27 13:53:59.000591, numSat: 6, Fix, Lat=35.681146, Lon=139.761146
numSatellites: 6
[ 0] Type:GPS, Id:12, Elv:55, Azm: 56, CN0:27.019999
[ 1] Type:GPS, Id:15, Elv:13, Azm:126, CN0:32.529999
[ 2] Type:GPS, Id:20, Elv:33, Azm:203, CN0:14.469999
[ 3] Type:GPS, Id:24, Elv:31, Azm: 62, CN0:30.179998
[ 4] Type:GPS, Id:29, Elv: 7, Azm:153, CN0:17.990000
[ 5] Type:QCA, Id:195, Elv:16, Azm:167, CN0:28.830000
2019/08/27 13:54:00.000581, numSat: 6, Fix, Lat=35.681147, Lon=139.761147
2019/08/27 13:54:01.000573, numSat: 6, Fix, Lat=35.681149, Lon=139.761149
2019/08/27 13:54:02.000574, numSat: 6, Fix, Lat=35.681148, Lon=139.761148
2019/08/27 13:54:03.000585, numSat: 6, Fix, Lat=35.681149, Lon=139.761149
2019/08/27 13:54:04.000577, numSat: 6, Fix, Lat=35.681153, Lon=139.761149
2019/08/27 13:54:05.000568, numSat: 6, Fix, Lat=35.681154, Lon=139.761149
2019/08/27 13:54:06.000589, numSat: 6, Fix, Lat=35.681154, Lon=139.761147
```

Autoscroll ☐ Show timestamp ☐ Newline ☐ 115200 baud ☐ Clear output

Overview of sketch

Declaration of header files, constants, objects, macros, functions

`void setup() {`

- Activate GNSS devices
- Select the satellite type
- Start positioning
- Declaration of methods printing time, position and number of satellite }

`void loop() {`

- Get and print the results of positioning (time, number of satellite, Id, latitude, longitude)
- stop and restart positioning every 5 minutes }

Sketch of GNSS (Global Positioning System)

```
#include <GNSS.h>
#define STRING_BUFFER_SIZE 128
#define RESTART_CYCLE (5*60)

static SpGnss Gnss; Gnss is object of SpGnss class

enum ParamSat {
    eSatGps,          /**< GPS           World wide coverage */
    eSatGlonass,       /**< GLONASS        World wide coverage */
    eSatGpsSbas,       /**< GPS+SBAS      North America */
    eSatGpsGlonass,    /**< GPS+Glonass   World wide coverage */
    eSatGpsBeidou,     /**< GPS+BeiDou    World wide coverage */
    eSatGpsGalileo,    /**< GPS+Galileo   World wide coverage */
    eSatGpsQz1c,       /**< GPS+QZSS_L1CA  East Asia & Oceania */
    eSatGpsGlonassQz1c, /**< GPS+Glonass+QZSS_L1CA East Asia & Oceania */
    eSatGpsBeidouQz1c, /**< GPS+BeiDou+QZSS_L1CA East Asia & Oceania */
    eSatGpsGalileoQz1c, /**< GPS+Galileo+QZSS_L1CA East Asia & Oceania */
    eSatGpsQz1cQz1S,  /**< GPS+QZSS_L1CA+QZSS_L1S Japan */
};

Select satellite

/* Set this parameter depending on your current region. */
static enum ParamSat satType = eSatGps;
```

```
static void Led_isActive(void) {
    static int state = 1;
    if (state == 1) {
        ledOn(PIN_LED0);
        state = 0;
    }
    else {
        ledOff(PIN_LED0);
        state = 1;
    }
}

static void Led_isPosfix(bool state) {
    if (state) {
        ledOn(PIN_LED1);
    }
    else {
        ledOff(PIN_LED1);
    }
}

static void Led_isError(bool state) {
    if (state) {
        ledOn(PIN_LED3);
    }
    else {
        ledOff(PIN_LED3);
    }
}
```

LED setting

Sketch of GNSS (Global Positioning System)

```
void setup() {  
  int error_flag = 0; /* put your setup code here, to run once: */  
  
  Serial.begin(115200); /* Set serial baudrate. */  
  
  sleep(3); /* Wait HW initialization done. */  
  
  ledOn(PIN_LED0); /* Turn on all LED: Setup start. */  
  ledOn(PIN_LED1);  
  ledOn(PIN_LED2);  
  ledOn(PIN_LED3);  
  
  Gnss.setDebugMode(PrintInfo);  
  
  int result;  
  
  result = Gnss.begin();  
}
```

Print debug messages about GNSS controlling and positioning if not set 0 to argument.

Activate GNSS device

```
if (result != 0) {  
  Serial.println("Gnss begin error!!");  
  error_flag = 1;  
}  
else {  
  switch (satType) {  
    case eSatGps:  
      Gnss.select(GPS);  
      break;  
  
    case eSatGpsSbas:  
      Gnss.select(GPS);  
      Gnss.select(SBAS);  
      break;  
  
    case eSatGlonass:  
      Gnss.select(GLONASS);  
      break;  
  
    case eSatGpsGlonass:  
      Gnss.select(GPS);  
      Gnss.select(GLONASS);  
      break;  
  
    case eSatGpsBeidou:  
      Gnss.select(GPS);  
      Gnss.select(BEIDOU);  
      break;  
  }
```

select(): Add specified satellite system to selection for positioning.

the types of satellite

```

case eSatGpsGalileo:
    Gnss.select(GPS);
    Gnss.select(GALILEO);
    break;

case eSatGpsQz1c:
    Gnss.select(GPS);
    Gnss.select(QZ_L1CA);
    break;

case eSatGpsQz1cQz1S:
    Gnss.select(GPS);
    Gnss.select(QZ_L1CA);
    Gnss.select(QZ_L1S);
    break;

case eSatGpsBeidouQz1c:
    Gnss.select(GPS);
    Gnss.select(BEIDOU);
    Gnss.select(QZ_L1CA);
    break;

case eSatGpsGalileoQz1c:
    Gnss.select(GPS);
    Gnss.select(GALILEO);
    Gnss.select(QZ_L1CA);
    break;

case eSatGpsGlonassQz1c:
default:
    Gnss.select(GPS);
    Gnss.select(GLONASS);
    Gnss.select(QZ_L1CA);
    break;
}

```

the types of satellite

```

result = Gnss.start(COLD_START);
if (result != 0) {
    Serial.println("Gnss start error!!");
    error_flag = 1;
}
else {
    Serial.println("Gnss setup OK");
}
}

ledOff(PIN_LED0); /* Turn off all LED:Setup done. */
ledOff(PIN_LED1);
ledOff(PIN_LED2);
ledOff(PIN_LED3);

/* Set error LED. */
if (error_flag == 1) {
    Led_isError(true);
    exit(0);
}
}

static void print_pos(SpNavData *pNavData) {
    char StringBuffer[STRING_BUFFER_SIZE];

    /* print time */
    snprintf(StringBuffer, STRING_BUFFER_SIZE, "%04d/%%02d/%%02d ", pNavData-
>time.year, pNavData->time.month, pNavData->time.day);
    Serial.print(StringBuffer);

    snprintf(StringBuffer, STRING_BUFFER_SIZE, "%02d:%%02d:%%02d.%06d, ",
pNavData->time.hour, pNavData->time.minute, pNavData->time.sec,
pNavData->time.usec);
    Serial.print(StringBuffer);
}

```

Start positioning

COLD_START: Discard all current position, time, and satellite orbital information. Start positioning from the beginning.

print_pos: print satellite positions

```

/* print satellites count */
snprintf(StringBuffer, STRING_BUFFER_SIZE, "numSat:%2d, ",
pNavData->numSatellites);
Serial.print(StringBuffer);

```

```

if (pNavData->posFixMode == FixInvalid) {
    Serial.print("No-Fix, "); /* print position data */
}
else {
    Serial.print("Fix, ");
}
if (pNavData->posDataExist == 0) {
    Serial.print("No Position");
}
else {
    Serial.print("Lat=");
    Serial.print(pNavData->latitude, 6);
    Serial.print(", Lon=");
    Serial.print(pNavData->longitude, 6);
}

```

```

Serial.println("");
}

```

numSatellites, posFixmode,
posDataExist, latitude, longitude
are public attributes of SpNavData
class

```

static void print_condition(SpNavData *pNavData) {
    char StringBuffer[STRING_BUFFER_SIZE];
    unsigned long cnt;

```

print_condition: print satellite
condition

```

    snprintf(StringBuffer, STRING_BUFFER_SIZE, "numSatellites:%2d¥n", pNavData->numSatellites);
    Serial.print(StringBuffer); /* Print satellite count. */

```

```

for (cnt = 0; cnt < pNavData->numSatellites; cnt++) {
    const char *pType = "---";
    SpSatelliteType sattype = pNavData->getSatelliteType(cnt);

```

```

    switch (sattype) {
        case GPS:
            pType = "GPS";
            break;

```

```

        case GLONASS:
            pType = "GLN";
            break;

```

```

        case QZ_L1CA:
            pType = "QCA";
            break;

```

```

        case SBAS:
            pType = "SBA";
            break;

```

```

        case QZ_L1S:
            pType = "Q1S";
            break;

```

```

        case BEIDOU:
            pType = "BDS";
            break;

```

```

        case GALILEO:
            pType = "GAL";
            break;

```

```

        default:
            pType = "UKN";
            break;
    }

```

Return types of all satellites
which was received

all satellite types


```

/* Get print conditions. */
unsigned long Id = pNavData->getSatelliteId(cnt);
unsigned long Elv = pNavData->getSatelliteElevation(cnt);
unsigned long Azm = pNavData->getSatelliteAzimuth(cnt);
float sigLevel = pNavData->getSatelliteSignalLevel(cnt);

/* Print satellite condition. */
snprintf(StringBuffer, STRING_BUFFER_SIZE, "[%2d] Type:%s, Id:%2d,
Elv:%2d, Azm:%3d, CN0:", cnt, pType, Id, Elv, Azm );
Serial.print(StringBuffer);
Serial.println(sigLevel, 6);
}
}

void loop() {
    static int LoopCount = 0;
    static int LastPrintMin = 0;

    Led_isActive(); /* Blink LED. */

    if (Gnss.waitUpdate(-1)) {
        SpNavData NavData;
        Gnss.getNavData(&NavData);

        /* Set posfix LED. */
        bool LedSet = (NavData.posDataExist && (NavData.posFixMode !=
FixInvalid));
        Led_isPosfix(LedSet);

        /* Print satellite information every minute. */
        if (NavData.time.minute != LastPrintMin) {
            print_condition(&NavData);
            LastPrintMin = NavData.time.minute;
        }
        /* Print position information. */
        print_pos(&NavData);
    }
}

```

Get the result of positioning

```

else {
    /* Not update. */
    Serial.println("data not update");
}

LoopCount++; /* Check loop count. */
if (LoopCount >= RESTART_CYCLE) {
    int error_flag = 0;
    ledOff(PIN_LED0); /* Turn off LED0 */
    Led_isPosfix(false); /* Set posfix LED. */

    /* Restart GNSS. */
    if (Gnss.stop() != 0) {
        Serial.println("Gnss stop error!!");
        error_flag = 1;
    }
    else if (Gnss.end() != 0) {
        Serial.println("Gnss end error!!");
        error_flag = 1;
    }
    else {
        Serial.println("Gnss stop OK.");
    }

    if (Gnss.begin() != 0) {
        Serial.println("Gnss begin error!!");
        error_flag = 1;
    }
    else if (Gnss.start(HOT_START) != 0) {
        Serial.println("Gnss start error!!");
        error_flag = 1;
    }
    else {
        Serial.println("Gnss restart OK.");
    }
    LoopCount = 0;
    if (error_flag == 1) { /* Set error LED. */
        Led_isError(true);
        exit(0);
    }
}
}
}

```

Positioning stop and restart
every 5 min
When you want to continue to
get the position data, you can
eliminate this loop process.

HOT_START: restart and get the
position information immediately
using the time, position and satellite
condition which was obtained just
before stop.


If you want to know further.
Please see the below site:
https://developer.sony.com/develop/spresense/docs/sdk_developer_guide_en.html#start_positioning

GNSS Library API Reference

- SpGnss class
 - <https://developer.sony.com/develop/spresense/developer-tools/api-reference/api-references-arduino/classSpGnss.html#ade8b8f5d5c05ba2ee67d857a32ef5229>
- SpNavData class
 - <https://developer.sony.com/develop/spresense/developer-tools/api-reference/api-references-arduino/classSpNavData.html#a29dd2b09d21ed4b7b60a19bce8de01b4>
- You can get the detailed information of each APIs used in this tutorial.
- Red letters are API.

Tutorial 2

- How to get the disaster information from QZSS in SPRESENSE**





- File → New

- Copy and paste the source code in the following page and upload

or

- Open QZSS.ino in Arduino IDE and upload



Overview of sketch

Declaration of header files, constants, objects, macros, functions

```
void setup() {
```

- Activate GNSS devices
- Select the satellite type }

```
void loop() {
```

- Get GNSS position data
- Print QZQSM sentences }

Sketch of QZSS

```
#include <GNSS.h>
#include <GNSSPositionData.h>
#include <gpsutils/cxd56_gnss_nmea.h>
```

```
SpGnss Gnss; Gnss is object of SpGnss class
char PositionData[sizeof(GnssPositionData)];
```

```
/* output NMEA */
static char nmea_buf[NMEA_SENTENCE_MAX_LEN];
```

```
static char *reqbuf(uint16_t size) {
    if (size > sizeof(nmea_buf)) {
        return NULL;
    }
    return nmea_buf;
}
```

Callback functions

```
static void freebuf(char *buf) {
    return;
}
```

```
static int outbin(char *buf, uint32_t len) {
    return len;
}
```

```
static int outnmea(char *buf) {
    return printf("%s", buf);
}
```

callback functions

```
void setup() {
    /* Initialize Serial */
    Serial.begin(115200);

    /* Initialize GNSS */
    if (Gnss.begin()) {
        Serial.println("begin error!");
    }

    /* select satellite system */
    Gnss.select(GPS); //Gnss.select(GLONASS);
    Gnss.select(QZ_L1CA);
    Gnss.select(QZ_L1S); //Gnss.select(SBAS);

    /* set interval */
    Gnss.setInterval(1);

    if (Gnss.start(COLD_START)) {
        Serial.println("start error!");
    }
}
```

Sketch of QZSS


```
/* use NMEA library */
NMEA_InitMask();
NMEA_SetMask(0x4000); // only QZQSM
NMEA_OUTPUT_CB funcs;
funcs.bufReq = reqbuf;
funcs.out = outnmea;
funcs.outBin = outbin;
funcs.bufFree = freebuf;
NMEA_RegistOutputFunc(&funcs);
}

void loop()
{
  /* Check update. */
  if (Gnss.waitUpdate(1000)) {
    /* Output NMEA */
    Gnss.getPositionData(PositionData);
    NMEA_Output(&(((GnssPositionData*)PositionData)->Data));

    void *handle;
    if (handle = Gnss.getDCReport()) {
      NMEA_DcReport_Output(handle);
    }
  }
}
```

Get position data and
print the QZQSM sentences

You can see the QZQSM sentences in serial monitor



SONY


Tutorial 3

– Play the recorded sound



Set up


- We can use anything as recorder, smart phone, laptop and so on.
- Please recode the sound as MP3 file.
- Save the MP3 file in the micro SD card
- Insert SD card to LTE extension board
- Connect head-phone or speaker



Install MP3 Decoder to SPRESENSE

STEP 1

Select "mp3_dec_installer" and upload




STEP 2

Select install type and baud rate



STEP3

Check if the installation is done



Overview of sketch

Declaration of header files, constants, objects, parameters, methods

```
void setup() {
```

- Initialize SD card
- Start Audio system
- Set clock mode, output device
- Initialize player with sampling rate, the location of mp3 file, stereo or monaural
- Read audio data
- Set the volume and start to play }

```
void loop() {
```

- Read audio data until file ends
- Stop player }

Sketch of code

```
#include <SDHCI.h>
#include <Audio.h>
```

```
SDClass theSD;
AudioClass *theAudio;
```

```
File myFile;
bool ErrEnd = false;
```

```
static void audio_attention_cb(const ErrorAttentionParam *atprm) {
    puts("Attention!");
```

```
    if (atprm->error_code >= AS_ATTENTION_CODE_WARNING) {
        ErrEnd = true;
    }
}
```

```
void setup() {
    while (!theSD.begin()) {
        Serial.println("Insert SD card."); /* wait until SD card is mounted. */
    }
}
```

begin(): Initialize the audio library and HW modules

theAudio is instance of AudioClass

```
theAudio = AudioClass::getInstance(); // start audio system
theAudio->begin(audio_attention_cb);
```

```
puts("initialization Audio Library");
```

Set clock mode to normal

```
theAudio->setRenderingClockMode(AS_CLKMODE_NORMAL);
theAudio->setPlayerMode(AS_SETPLAYER_OUTPUTDEVICE_SPHP,
AS_SP_DRV_MODE_LINEOUT);
```

In case of I2S
AS_SETPLAYER_OUTPUTDEVICE_I2SOUTPUT

Initialize player

location of mp3 decoder

```
err_t err = theAudio->initPlayer(AudioClass::Player0, AS_CODECTYPE_MP3,
"/mnt/sd0/BIN", AS_SAMPLINGRATE_AUTO, AS_CHANNEL_STEREO);
```

In case of monaural
AS_CHANNEL_MONO

Only mp3 allows to set AUTO

```
/* Verify player initialize */
if (err != AUDIOLIB_ECODE_OK) {
    printf("Player0 initialize error\n");
    exit(1);
}
```

```
myFile = theSD.open("Sine.mp3"); /* Open file placed on SD card */
```

```
/* Verify file open */
if (!myFile) {
    printf("File open error\n");
    exit(1);
}
printf("Open! %d\n", myFile);
```

Read audio data from SD card
and write it to FIFO

```
/* Send first frames to be decoded */
err = theAudio->writeFrames(AudioClass::Player0, myFile);
```

```
if ((err != AUDIOLIB_ECODE_OK) && (err != AUDIOLIB_ECODE_FILEEND)){
    printf("File Read Error! =%d\n", err);
    myFile.close();
    exit(1);
}
```

```
puts("Play!");
```

Start to play

```
theAudio->setVolume(30); /* Main volume [dB] set */
theAudio->startPlayer(AudioClass::Player0);
}
```

Sketch of code

```
void loop() {
  puts("loop!!");

  /* Send new frames to decode in a loop until file ends */
  int err = theAudio->writeFrames(AudioClass::Player0, myFile);

  /* Tell when player file ends */
  if (err == AUDIOLIB_ECODE_FILEEND) {
    printf("Main player File End!¥n");
  }

  /* Show error code from player and stop */
  if (err) {
    printf("Main player error code: %d¥n", err);
    goto stop_player;
  }

  if (ErrEnd) {
    printf("Error End¥n");
    goto stop_player;
  }

  /* This sleep is adjusted by the time to read the audio stream file.
   Please adjust in according with the processing contents
   being processed at the same time by Application.
  */
  usleep(40000);
```

Read audio data from SD card
and write it to FIFO

```
/* Don't go further and continue play */
return;
```

```
stop_player:
  sleep(1);
  theAudio->stopPlayer(AudioClass::Player0);
  myFile.close();
  exit(1);
}
```

Stop Audio

Audio Library API Reference

- <https://developer.sony.com/develop/spresense/developer-tools/api-reference/api-references-arduino/classAudioClass.html#ae740705f6a3e40a94546fc4e70950925>
- You can get the detailed information of each APIs used in this tutorial.
- Red letters are API

SONY

SONY is a registered trademark of Sony Corporation.

Names of Sony products and services are the registered trademarks and/or trademarks of Sony Corporation or its Group companies.

Other company names and product names are registered trademarks and/or trademarks of the respective companies.